

## 17 Using Laplace Transforms

This Maple V session illustrates how Laplace transforms can be used to solve linear differential equations with constant coefficients. Laplace transforms are especially useful in solving equations in which discontinuities appear. Using built-in Maple V Libraries allows one to include the unit step or Heaviside Function as well as an infinite impulse or Dirac Delta Function. Maple commands that may be new to you for this session include:

We first show how the **laplace** option for **dsolve** can be used to solve the d.e.

$$y'' + 3y' - y = \cos(t).$$

As always, we must first enter the equation.

```
> deq := diff(y(t), t$2) + 3*diff(y(t), t) - y(t) = cos(t);
```

$$deq := \left( \frac{\partial^2}{\partial t^2} y(t) \right) + 3 \left( \frac{\partial}{\partial t} y(t) \right) - y(t) = \cos(t)$$

We now use **dsolve** with the **laplace** option to solve the differential equation.

```
> dsolve(deq, y(t), laplace);
```

$$\begin{aligned} y(t) = & \frac{3}{13} \sin(t) - \frac{2}{13} \cos(t) + \frac{3}{13} y(0) e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} \\ & + \frac{2}{13} D(y)(0) e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} + y(0) e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \\ & + \frac{2}{13} e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \end{aligned}$$

Notice that the output contains  $y(0)$  and  $D(y)(0)$  as constants. If initial conditions for this method are specified they must be at 0. Solving this equation without the **laplace** option results in a very different looking answer since Maple V uses a different algorithm to solve the equation. It is difficult to verify that the solutions are the same, and sometimes even Maple V is not able to simplify the difference of the two answers to 0. In those cases you may conclude that both answers agree if you can verify that both answers are actually solutions, and the uniqueness theorem applies in this, the linear case, to prove that they are the same. As an example let us use the initial conditions:

$$y(0) = 1, y'(0) = 1.$$

```
> init := y(0)=1, D(y)(0)=1;
```

$$init := y(0) = 1, D(y)(0) = 1$$

The solution to the initial value problem using the **laplace** option yields:

```
> sol1 := dsolve({deq, init}, y(t), laplace);
```

$$\begin{aligned} \text{sol1} := y(t) = & \frac{3}{13} \sin(t) - \frac{2}{13} \cos(t) + \frac{5}{13} e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} \\ & + \frac{15}{13} e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \end{aligned}$$

The solution using **dsolve** with no options is

```
> sol2 := dsolve({deq, init}, y(t));
```

$$\begin{aligned} \text{sol2} := y(t) = & -\frac{2}{13} \cos(t) + \frac{3}{13} \sin(t) + \frac{\left(\frac{5}{26} \sqrt{13} + \frac{15}{26}\right) \sqrt{e^{(\sqrt{13}t)}}}{(e^t)^{3/2}} + \frac{-\frac{5}{26} \sqrt{13} + \frac{15}{26}}{(e^t)^{3/2} \sqrt{e^{(\sqrt{13}t)}}} \end{aligned}$$

You might feel that these two solutions are not equal. However, the next three Maple commands demonstrate their equality.

```
> expr1 := expand(convert(sol1, exp));
```

$$\begin{aligned} \text{expr1} := y(t) = & -\frac{3}{26} I e^{(It)} + \frac{3}{26} \frac{I}{e^{(It)}} - \frac{1}{13} e^{(It)} - \frac{1}{13} \frac{1}{e^{(It)}} + \frac{5}{26} \frac{\sqrt{13} \sqrt{e^{(\sqrt{13}t)}}}{(e^t)^{3/2}} \\ & - \frac{5}{26} \frac{\sqrt{13}}{(e^t)^{3/2} \sqrt{e^{(\sqrt{13}t)}}} + \frac{15}{26} \frac{\sqrt{e^{(\sqrt{13}t)}}}{(e^t)^{3/2}} + \frac{15}{26} \frac{1}{(e^t)^{3/2} \sqrt{e^{(\sqrt{13}t)}}} \end{aligned}$$

```
> expr2 := expand(convert(sol2, exp));
```

$$\begin{aligned} \text{expr2} := y(t) = & -\frac{3}{26} I e^{(It)} + \frac{3}{26} \frac{I}{e^{(It)}} - \frac{1}{13} e^{(It)} - \frac{1}{13} \frac{1}{e^{(It)}} + \frac{5}{26} \frac{\sqrt{13} \sqrt{e^{(\sqrt{13}t)}}}{(e^t)^{3/2}} \\ & - \frac{5}{26} \frac{\sqrt{13}}{(e^t)^{3/2} \sqrt{e^{(\sqrt{13}t)}}} + \frac{15}{26} \frac{\sqrt{e^{(\sqrt{13}t)}}}{(e^t)^{3/2}} + \frac{15}{26} \frac{1}{(e^t)^{3/2} \sqrt{e^{(\sqrt{13}t)}}} \end{aligned}$$

```
> expr1 - expr2;
```

$$0 = 0$$

Notice that if you simply subtract sol2 from sol1 without the simplification commands that you get a very complicated expression that is not obviously zero.

```
> simplify(sol1 - sol2);
```

$$\begin{aligned} 0 = & \frac{5}{13} e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} + \frac{15}{13} e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \\ & - \frac{5}{26} e^{(-3/2t+1/2\sqrt{13}t)} \sqrt{13} - \frac{15}{26} e^{(-3/2t+1/2\sqrt{13}t)} + \frac{5}{26} e^{(-3/2t-1/2\sqrt{13}t)} \sqrt{13} \\ & - \frac{15}{26} e^{(-3/2t-1/2\sqrt{13}t)} \end{aligned}$$

But,

```
> expand(convert(sol1-sol2,exp));
```

$$0 = 0$$

Thus, we conclude that in this case we that the solutions that are obtained by the two different methods are identical.

The **laplace** option for **dsolve** is based on computing Laplace transforms. Maple has a built-in procedure for computing Laplace transforms called **laplace**. To use it to solve an equation the following steps are used.

1. Take the Laplace transform of the entire equation.
2. Solve for the Laplace transform of the dependent variable.
3. Apply the inverse Laplace transform to the resulting equation.

We now use this method to solve the above initial value problem.

1. Take the Laplace transform the entire equation.

```
> Leq := laplace(deq,t,s);
```

$$\begin{aligned} Leq := & (\text{laplace}(y(t), t, s) s - y(0)) s - D(y)(0) + 3 \text{laplace}(y(t), t, s) s - 3 y(0) \\ & - \text{laplace}(y(t), t, s) = \frac{s}{s^2 + 1} \end{aligned}$$

2. Solve for the Laplace transform of the dependent variable.

```
> LY := solve(Leq ,laplace(y(t),t,s));
```

$$LY := - \frac{-s y(0) - D(y)(0) - 3 y(0) - \frac{s}{s^2 + 1}}{s^2 + 3 s - 1}$$

3. Apply the Inverse Laplace transform to the resulting expression.

```
> sol := invlaplace(LY,s,t);
```

$$\begin{aligned} sol := & \frac{3}{13} \sin(t) - \frac{2}{13} \cos(t) + \frac{3}{13} y(0) e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} \\ & + \frac{2}{13} D(y)(0) e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} + y(0) e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \\ & + \frac{2}{13} e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right) \end{aligned}$$

Now substitute the initial conditions.

```
> subs({y(0)=1,D(y)(0)=1},sol);
```

$$\frac{3}{13} \sin(t) - \frac{2}{13} \cos(t) + \frac{5}{13} e^{(-3/2)t} \sinh\left(\frac{1}{2} \sqrt{13} t\right) \sqrt{13} + \frac{15}{13} e^{(-3/2)t} \cosh\left(\frac{1}{2} \sqrt{13} t\right)$$

The above answer should agree with sol1. In fact

```
> simplify(" - rhs(sol1));
```

0

The method of Laplace transforms is especially applicable in cases where the function on the right hand side of the d.e. is discontinuous. Many functions with finite jump discontinuities can be expressed in terms of a simple function known as the Heaviside function,  $H(t)$ :

$$H(t) = 0 \quad \text{if } t < 0 \quad \text{and} \quad H(t) = 1 \quad \text{if } t \geq 0.$$

Maple has a the unit step function **Heaviside**.

We now solve the d.e.

$$y'' + 5y' + 6y = f(t), \quad y(0) = 0, \quad y'(0) = 1$$

where

$$f(t) = 1 \quad \text{if } 0 \leq t \leq 1, \quad \text{and} \quad f(t) = 0 \quad \text{if } t > 1.$$

```
> deq2 := diff(y(t), t$2) + 5*diff(y(t), t) + 6*y(t) =
Heaviside(t)-Heaviside(t-1);
```

$$deq2 := \left( \frac{\partial^2}{\partial t^2} y(t) \right) + 5 \left( \frac{\partial}{\partial t} y(t) \right) + 6y(t) = \text{Heaviside}(t) - \text{Heaviside}(t-1)$$

```
> y2 := rhs(dsolve({deq2, y(0)=0, D(y)(0)=1}, y(t), laplace));
```

$$y2 := -\frac{2}{3} e^{(-3t)} + \frac{1}{2} e^{(-2t)} + \frac{1}{6} - \text{Heaviside}(t-1) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+3)} - \frac{1}{2} e^{(-2t+2)} \right)$$

We now verify that y2 is the solution. First the initial conditions:

```
> simplify(subs(t=0, y2));
```

0

```
> simplify(subs(t=0, diff(y2, t)));
```

1

Now we check if y2 satisfies the differential equation.


```
simplify(subs(y(t)=y2, deq2));
```

$$1 - \text{Heaviside}(t-1) = \text{Heaviside}(t) - \text{Heaviside}(t-1)$$

Note, that since the Heaviside function is equal to 1 for  $t \geq 0$ , the above equation is true for all positive  $t$ .

We can also plot the graph of the solution:

```
> plot(y2,t= 0..10);
```




More complicated combinations can also be solved.

```
> f := sum(Heaviside(t-2*n)-Heaviside(t-2*n-1),n=0..4);
```

$$\begin{aligned} f := & \text{Heaviside}(t) - \text{Heaviside}(t-1) + \text{Heaviside}(t-2) - \text{Heaviside}(t-3) \\ & + \text{Heaviside}(t-4) - \text{Heaviside}(t-5) + \text{Heaviside}(t-6) - \text{Heaviside}(t-7) \\ & + \text{Heaviside}(t-8) - \text{Heaviside}(t-9) \end{aligned}$$

Before finding a solution of this equation, we give a plot of its right hand side.

```
> plot(f(t),t=0..10);
```



```
> deq3 := diff(y(t),t$2) + 5*diff(y(t),t) + 6*y(t) = f;
```

$$\begin{aligned} deq3 := & \left( \frac{\partial^2}{\partial t^2} y(t) \right) + 5 \left( \frac{\partial}{\partial t} y(t) \right) + 6y(t) = \text{Heaviside}(t) - \text{Heaviside}(t-1) \\ & + \text{Heaviside}(t-2) - \text{Heaviside}(t-3) + \text{Heaviside}(t-4) - \text{Heaviside}(t-5) \\ & + \text{Heaviside}(t-6) - \text{Heaviside}(t-7) + \text{Heaviside}(t-8) - \text{Heaviside}(t-9) \end{aligned}$$


```
> y3 := rhs(dsolve({deq3,y(0)=0,D(y)(0)=1},y(t),laplace));
```

$$\begin{aligned} y3 := & -\frac{2}{3}e^{(-3t)} + \frac{1}{2}e^{(-2t)} + \frac{1}{6} - \text{Heaviside}(t-1) \left( \frac{1}{6} + \frac{1}{3}e^{(-3t+3)} - \frac{1}{2}e^{(-2t+2)} \right) \\ & + \text{Heaviside}(t-2) \left( \frac{1}{6} + \frac{1}{3}e^{(-3t+6)} - \frac{1}{2}e^{(-2t+4)} \right) \end{aligned}$$

$$\begin{aligned}
& - \text{Heaviside}(t-3) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+9)} - \frac{1}{2} e^{(-2t+6)} \right) \\
& + \text{Heaviside}(t-4) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+12)} - \frac{1}{2} e^{(-2t+8)} \right) \\
& - \text{Heaviside}(t-5) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+15)} - \frac{1}{2} e^{(-2t+10)} \right) \\
& + \text{Heaviside}(t-6) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+18)} - \frac{1}{2} e^{(-2t+12)} \right) \\
& - \text{Heaviside}(t-7) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+21)} - \frac{1}{2} e^{(-2t+14)} \right) \\
& + \text{Heaviside}(t-8) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+24)} - \frac{1}{2} e^{(-2t+16)} \right) \\
& - \text{Heaviside}(t-9) \left( \frac{1}{6} + \frac{1}{3} e^{(-3t+27)} - \frac{1}{2} e^{(-2t+18)} \right)
\end{aligned}$$

The graph of this solution is interesting.

```
> plot(y3,t=0..10);
```



We now use **dsolve** with the **laplace** option to solve equations that involve the Dirac delta function which emulates an infinite impulse of 0 duration.

```
> d := sum(Dirac(t-2*n*Pi), n=0..4);
```

$$d := \text{Dirac}(t) + \text{Dirac}(t - 2\pi) + \text{Dirac}(t - 4\pi) + \text{Dirac}(t - 6\pi) + \text{Dirac}(t - 8\pi)$$

We use the above sum of Dirac Delta Functions as the right hand side of the following equation.

```
> deq := diff(y(t), t$2) + 4*y(t) = d;
```

$$\begin{aligned}
deq := & \left( \frac{\partial^2}{\partial t^2} y(t) \right) + 4y(t) = \\
& \text{Dirac}(t) + \text{Dirac}(t - 2\pi) + \text{Dirac}(t - 4\pi) + \text{Dirac}(t - 6\pi) + \text{Dirac}(t - 8\pi)
\end{aligned}$$

```
> init := y(0)=0, D(y)(0)=0;
```

$$init := y(0) = 0, D(y)(0) = 0$$

```
> y3 := rhs(dsolve({deq,init},y(t),laplace));
```

$$\begin{aligned} y3 := & \frac{1}{2} \sin(2t) + \frac{1}{2} \text{Heaviside}(t - 2\pi) \sin(2t - 4\pi) \\ & + \frac{1}{2} \text{Heaviside}(t - 4\pi) \sin(2t - 8\pi) + \frac{1}{2} \text{Heaviside}(t - 6\pi) \sin(2t - 12\pi) \\ & + \frac{1}{2} \text{Heaviside}(t - 8\pi) \sin(2t - 16\pi) \end{aligned}$$


This function satisfies the differential equation since:

```
> simplify(subs(y(t)=y3,deq));
```

$$\begin{aligned} \text{Dirac}(t - 2\pi) + \text{Dirac}(t - 4\pi) + \text{Dirac}(t - 6\pi) + \text{Dirac}(t - 8\pi) = \\ \text{Dirac}(t) + \text{Dirac}(t - 2\pi) + \text{Dirac}(t - 4\pi) + \text{Dirac}(t - 6\pi) + \text{Dirac}(t - 8\pi) \end{aligned}$$

We now plot the last solution:

```
> plot(y3,t=0..10);
```



**Exercises 17.0** In this problem you are asked to solve the following initial value problem:

$$L(y)(t) = y''(t) + \omega^2 y(t) = f(t), \quad y(0) = a, \quad y'(0) = b,$$

for various values of  $\omega$ ,  $a$ , and  $b$ , and different functions  $f(t)$ .

1. Use **dsolve** with the **laplace** option to solve:

$$y''(t) + 25y(t) = \sin(t), \quad y(0) = 1, \quad y'(0) = 1/2$$

2. Plot the graph of the solution to problem 1 on the interval on the interval  $[0, \pi]$ .

3. Use **dsolve** with the **laplace** option to solve:

$$y''(t) + 25y(t) = \sin(5t), \quad y(0) = 1, \quad y'(0) = 1/2$$

4. Plot the graph of the solution to problem 3 on the interval on the interval  $[0, 4\pi]$ .

5. Use Maple V and the built in **Heaviside** function to express the function

$$f(t) = 2 \quad \text{if } 0 \leq t < \pi/2 \quad \text{and} \quad f(t) = 1 \quad \text{if } t \geq \pi/2.$$

in a closed form.

6. Use **dsolve** with the **laplace** option to solve

$$y''(t) + 25y(t) = f(t), \quad y(0) = 2, \quad y'(0) = 1$$

where

$$f(t) = 2 \quad \text{if } 0 \leq t < \pi/2 \quad \text{and} \quad f(t) = 1 \quad \text{if } t \geq \pi/2.$$

7. Plot the graph of the solution to problem 6 on the interval on the interval  $[0, \pi]$ . Show the command you use and paste the graph at the prompts listed below:
8. Use Maple and the built in **Heaviside** function to express the function

$$g(t) = t^2 \quad \text{if } 0 \leq t < 4 \quad \text{and} \quad g(t) = 4 \quad \text{if } t \geq 4$$

in a closed form.

9. Use **dsolve** with the **laplace** option to solve:

$$y''(t) + 25y(t) = g(t), \quad y(0) = 0, \quad y'(0) = 0$$

where

$$g(t) = t^2 \quad \text{if } 0 \leq t < 4 \quad \text{and} \quad g(t) = 4 \quad \text{if } t \geq 4$$

10. Plot the graph of the solution to problem 9 on the interval on the interval  $[0, 6]$ .