

## 19 Fourier Series

This Maple V segment illustrates how Maple V can be utilized to study Fourier Series. Numerical and graphical methods for estimating the error of the approximation for truncating the series to  $n$  terms are suggested. An example of the Gibb's Phenomenon is discussed.

Suppose a function,  $F(x)$ , is defined and integrable on an interval  $[-L, L]$ . Then  $F(x)$  has a Fourier Expansion given by the formula

$$F(x) = \frac{1}{2}a_0 + \left( \sum_{n=1}^{\infty} \left( a_n \cos\left(\frac{n\pi x}{L}\right) + b_n \sin\left(\frac{n\pi x}{L}\right) \right) \right)$$

where for  $n=0,1,2,\dots$ ,

$$a_n = \frac{\int_{-L}^L F(x) \cos\left(\frac{n\pi x}{L}\right) dx}{L}$$

and

$$b_n = \frac{\int_{-L}^L F(x) \sin\left(\frac{n\pi x}{L}\right) dx}{L}$$

If the function  $F$  is even or odd then the series can be simplified. For example if  $F(x)$  is even, *i.e.*,  $F(x) = F(-x)$  then the series reduces to the Cosine series

$$F(x) = \frac{1}{2}a_0 + \left( \sum_{n=1}^{\infty} a_n \cos\left(\frac{n\pi x}{L}\right) \right)$$

where

$$a_n = 2 \frac{\int_0^L F(x) \cos\left(\frac{n\pi x}{L}\right) dx}{L}$$

If  $F(x)$  is odd, *i.e.*, if  $F(-x) = -F(x)$ , then the Fourier Series reduces to the Sine series

$$F(x) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi x}{L}\right)$$

where

$$b_n = 2 \frac{\int_0^L F(x) \sin\left(\frac{n\pi x}{L}\right) dx}{L}$$

Our first example, a “saw tooth” function, is the odd extension of the following function, resulting in a function of period  $2\pi$ .

$$F(x) = x, \quad \text{if } 0 \leq x \leq \pi/2,$$

and

$$F(x) = \pi - x, \quad \text{for } \pi/2 \leq x \leq \pi.$$

This function can be defined (at least for the interval  $[-\pi, \pi]$ ) by means of the Heaviside Function. When using this function with Maple V it is convenient to abbreviate the built-in **Heaviside** Function with the **alias**, **H**.

```
> alias(H=Heaviside);
```

$I, H$


The desired function (in the interval  $[-\pi, \pi]$ ) can now be defined.

```
> F := x -> -(Pi+x)*(H(x+3*Pi/2)-H(x+Pi/2))+x*(H(x+Pi/2)-H(x-Pi/2))+
  (Pi-x)*H(x-Pi/2);
```

$$F := x \rightarrow -(\pi + x) \left( H\left(x + \frac{3}{2}\pi\right) - H\left(x + \frac{1}{2}\pi\right) \right) + x \left( H\left(x + \frac{1}{2}\pi\right) - H\left(x - \frac{1}{2}\pi\right) \right) \\ + (\pi - x) H\left(x - \frac{1}{2}\pi\right)$$

We now give a plot of the function  $F(x)$  on the interval  $[-\pi, \pi]$ . Note the use of the punctuation symbols!

```
> Plot1 := plot(F(x), x=-Pi..Pi):
```



Observe that since  $F(x)$  is odd that the Fourier expansion consists only of sines. Thus we need only compute, the  $b$ 's. The appropriate Maple V command is

```
> b := n -> (2/Pi)*(int(x*sin(n*x), x=0..Pi/2) + int((Pi-x)*sin(n*x),
  x=Pi/2..Pi));
```

$$b := n \rightarrow 2 \frac{\int_0^{1/2\pi} x \sin(nx) dx + \int_{1/2\pi}^{\pi} (\pi - x) \sin(nx) dx}{\pi}$$

A few values of  $b$  can be written to see if things look correct.

```
> b(1); b(2); b(3); b(4); b(5);
```

$$4 \frac{1}{\pi}$$

$$0$$

$$-\frac{4}{9} \frac{1}{\pi}$$

$$0$$

$$\frac{4}{25} \frac{1}{\pi}$$


Now we can write the first  $n$  terms of the Fourier expansion.

```
> s := (n,x) -> sum(b(k)*sin(k*x),k=1..n);
```

$$s := (n, x) \rightarrow \sum_{k=1}^n b(k) \sin(kx)$$

A plot of the first 15 terms of this expansion is given below.

```
plot(s(15,x),x=-Pi..Pi);
```




One way to see how well the first 15 terms of the Fourier expansion for  $F$  approximates  $F$  is to compute the corresponding error function or the difference between the approximation and the original function.

```
> e := (n,x)-> s(n,x) - F(x);
```

$$e := (n, x) \rightarrow s(n, x) - F(x)$$

```
> plot(e(15,x),x=-Pi..Pi);
```




**Note:** At first glance the error seems to be large at  $\pi/2$ , but note that the value that it actually has is small. By using the mouse to move to near the point with x coordinate  $\pi/2$  at the bottom of the graph of  $e(15,x)$  and clicking once we can estimate that

$$|e(15, x)| < .04$$


for all  $x$  in the interval  $[-\pi, \pi]$ .

```
> plot(e(31,x), x=-Pi..Pi);
```




This time, using the mouse as above, we estimate the error to be bounded by .021.

```
> plot(e(51,x), x =-Pi..Pi);
```



This time, using the mouse as above, we estimate the error to be bounded by .015.

```
> plot({e(15,x), e(31,x), e(51,x)}, x=-Pi..Pi);
```



By observing the error function for different values of  $n$  it seems safe to conjecture that the error in the approximation,  $e(n,x)$ , becomes small uniformly in  $x$  as  $n$  gets large.

### Animation Demonstration:

Let us now make an animation that provides graphical evidence of how well the finite sums approximate the function  $F(x)$ . First we make a plot of the function  $F(x)$  to use later. The following loop develops six frames that show a plot of  $F(x)$  along with  $s(n,x)$  on the same graph for values of  $n$  equal to

$$1, 3, 5, \dots, 11.$$

```
> i := 'i';

                                     i := i

> for i from 0 to 5 do
> m := 2*i+1;
> R := plots[textplot]([-1.5,1,'n = '.m]):
> Q := plot(s(m,x),x=-Pi..Pi):
> P[i] := plots[display]({Plot1, R, Q}):
> od:
> i := 'i';
```

$$i := i$$

The next command creates an animation with 6 frames.

```
> plots[display]([seq(plot(P[i]),i=0..5)],insequence=true);
```

The output of the last Maple V instruction should be an animated sequence of graphs that compares the  $n$ th approximation,  $s(n,x)$ , to  $F(x)$  for six different values of  $x$ .

The next set of animations compares the error made in each approximation with  $F(x)$ .

```

> i := 'i';

                                i := i

> for i from 0 to 5 do
> m := 2*i+1;
> R := plots[textplot]([-1.5,1,'n = '.m]):
Q := plot(e(m,x),x=-Pi..Pi):
> P[i] := plots[display]({Plot1, R, Q}):
od:
> i := 'i';

                                i := i

> plots[display]([seq(P[i],i=0..5)],insequence=true);

```

The preceding Maple V output is a sequence of plots of the the error function,  $e(n,x)$ , along with the original function  $F(x)$ . It should give you you some idea of how well the  $n$ th approximation approximates  $F(x)$ .

Let us now take the even periodic extension of the above function,  $F(x)$ . We can again use the Heaviside Function to define this extension in the interval  $[-\pi, \pi]$ . We will call this function  $G(x)$ .

```

> G := (x+Pi)*(H(x+Pi)-H(x+Pi/2)) - x*(H(x+Pi/2)-H(x)) +
      x*(H(x)-H(x-Pi/2))+(Pi-x)*H(x-Pi/2);

```


$$\begin{aligned}
 G := (x + \pi) & \left( H(x + \pi) - H\left(x + \frac{1}{2}\pi\right) \right) - x \left( H\left(x + \frac{1}{2}\pi\right) - H(x) \right) \\
 & + x \left( H(x) - H\left(x - \frac{1}{2}\pi\right) \right) + (\pi - x) H\left(x - \frac{1}{2}\pi\right)
 \end{aligned}$$

We can plot the function just to make sure that it is the desired function.

```

> plot(G,x=-Pi..Pi);

```



Since  $G$  is an even function we must approximate it with a cosine series. First we compute the  $a$ 's.

```
> a := n -> (2/Pi)*(int(x*cos(n*x), x=0..Pi/2)+int((Pi-x)*cos(n*x),
    x=Pi/2..Pi));
```

$$a := n \rightarrow 2 \frac{\int_0^{1/2\pi} x \cos(nx) dx + \int_{1/2\pi}^{\pi} (\pi - x) \cos(nx) dx}{\pi}$$

Note that this is analogous to the formula for the b's except cosines replace sines. Evaluate a few of the terms:

```
> a(0); a(1); a(2);
```

$$\frac{1}{2}\pi$$

$$0$$

$$-2\frac{1}{\pi}$$

Now we can compute a Fourier Cosine Sum.

```
> s := (n,x) -> a(0)/2 + sum(a(k)*cos(k*x), k=1..n);
```

$$s := (n, x) \rightarrow \frac{1}{2}a(0) + \left( \sum_{k=1}^n a(k) \cos(kx) \right)$$

It is instructive to plot  $s(15, x)$  and  $G(x)$  on the same graph. One can make a similar analysis as above concerning estimating the error of approximation when replacing  $G(x)$  by  $s(n, x)$  for some values of  $n$ .

Sometimes there is a surprise in store for us as we try to obtain error estimates. In case the function  $x$  has a jump discontinuity we can observe an unusual phenomenon. In this example we consider the odd periodic extension of period  $2\pi$  of the function:

$$F1(x) = \pi - x, \quad 0 \leq x \leq \pi.$$

**Note:** This function has jump discontinuities at

$$\dots - \pi, 0, \pi, \dots$$


We can obtain, by using the Heaviside Function, a closed form for the function  $F1$  in the interval  $[-2\pi, 2\pi]$ .

```
> F1 := x -> (Pi-x)*H(x)/2-(Pi+x)*H(-x)/2;
```

$$F1 := x \rightarrow \frac{1}{2}(\pi - x)H(x) - \frac{1}{2}(\pi + x)H(-x)$$

```
> Plot1 :=
```

```
plot(F1(x), x=-2*Pi..2*Pi): "
```



```
> b := n -> (2/Pi)*int((Pi-x)/2*sin(n*x), x=0..Pi);
```

$$b := n \rightarrow 2 \frac{\int_0^\pi \frac{1}{2} (\pi - x) \sin(nx) dx}{\pi}$$

```
> b(1);b(2);b(3);
```

$$1$$


$$\frac{1}{2}$$

$$\frac{1}{3}$$

```
> s := (n,x) -> sum(b(k)*sin(k*x), k=1..n);
```

$$s := (n, x) \rightarrow \sum_{k=1}^n b(k) \sin(kx)$$

```
> plot({s(9,x), f1(x)}, x=-2*Pi..2*Pi);
```




In this case there is a noticeable “hump” in the curve of  $s(9,x)$  near the discontinuities at  $0$ ,  $-2\pi$ , and  $2\pi$ . To analyze this situation let us examine the error function.

```
> e := (n,x) -> s(n,x) - F1(x);
```


$$e := (n, x) \rightarrow s(n, x) - F1(x)$$

```
> plot(e(9,x),x=0..1);
```




We see that the error term takes on a value nearly equal to  $-\pi/2$  at  $x = 0$ . This is not surprising since, at the discontinuity, theory suggests that the Fourier Series converges to the value  $(F(0_-) + F(0_+))/2$  which is 0 in this case since  $F1(0_-) = -\pi/2$  and  $F1(0_+) = \pi/2$ . The thing to observe here is the size of the “hump” which is located near .31. Here we see the error is nearly equal to .28.

```
> plot(e(15,x),x=0..1);
```




```
> plot(e(44,x),x=0..1);
```



The “hump” of approximately the same


size is still present even for this much larger value of  $n$ .

```
> plot( {e(9,x),e(15,x),e(44,x)},x=0..1);
```



Note that each of the curves  $e(9,x)$ ,  $e(15,x)$ , and  $e(44,x)$  oscillate and each have a maximum value of about .28. It seems that this maximum value is nearly .28 no matter what value  $n$  assumes. For example:

```
> plot( {e(44,x),e(70,x)},x=0..1,y=0..(0.3));
```



This as an example of what is known as Gibbs’s Phenomenon and it can be shown that the actual value of the maximum approaches the value of the integral:

$$\int_0^{\pi} \frac{\sin(z)}{z} dz - \frac{1}{2}\pi$$

```
> evalf(int(sin(z)/z, z=0..Pi))-evalf(Pi/2);
```

```
.281140725
```

Thus it can be proven (This is not easy.) that the maximum peak of  $e(n, x)$  is around .28 for all  $n$ .

### Animation Demonstration:

Let us now make an animation that gives further graphical amplification to these ideas on the Gibb's Phenomenon. As before we develop six frames each of which is a plot of the finite sum,  $s(n, x)$ , along with  $F(x)$  for values of  $n = 1, 5, 9, \dots, 21$ .

```
> i := 'i';
```

```
i := i
```

```
> for i from 0 to 5 do
```

```
> m := 4*i+1;
```

```
> R := plots[textplot]([-3.0, 1, 'n = ' . m]):
```

```
> Q := plot(s(m, x), x=-2*Pi..2*Pi):
```

```
> P[i] := plots[display]({Plot1, R, Q}):
```

```
od:
```

```
> i := 'i';
```

```
i := i
```

```
> plots[display]([seq(P[i], i=0..5)], insequence=true);
```

We finish the animation analysis with the following Maple V segment, which produces six frames of the error function,  $e(n, x)$ , along with the function  $F(x)$ , for the values  $n = 1, 5, 9, \dots, 21$ .

```
> i := 'i';
```

```
i := i
```

```
> for i from 0 to 5 do
```

```
> m := 4*i+1;
```

```
> R := plots[textplot]([-3.0, 1, 'n = ' . m]):
```

```
> Q := plot(e(m,x),x=-2*Pi..2*Pi):
```

```
> P[i] := plots[display]({Plot1, R, Q}):
```

```
> od:
```


```
> i := 'i';
```

$$i := i$$

```
> plots[display]([seq(P[i],i=0..5)],insequence=true);
```


If we stay away from points of discontinuity then convergence is again uniform. The following sequence of commands give an indication that the error gets smaller with large  $n$  over the interval  $[0.5,1]$ .

```
> plot(e(9,x),x=.5..1);
```




In This case  $e(9,x)$  takes on a maximum value of around 0.156.

```
> plot(e(44,x),x=.5..1);
```




Here the error seems to be bounded by about 0.043.

```
plot(e(70,x),x=.5..1);
```



Here the maximum error is bounded by 0.027. One might feel that there could be a “hump” that is simply smaller than the one found above. But if we plot more points then the error continues to diminish. If we taken = 100 then

```
plot(e(100,x),x=.5..1);
```



The error in this case seems bounded by 0.02.

### Exercises 19.0

1. Using the Heaviside Function write a Maple V statement that yields the even  $2\pi$  periodic extension of the function:

$$G1(x) = \pi - x, \quad 0 \leq x \leq \pi$$

valid over the interval  $[-2\pi, 2\pi]$ . **Note:** This means that your plot will extend over two periods of the function.

2. Write a Maple V function that defines the “a” coefficients in the Fourier Cosine expansion of  $G1(x)$ .
3. Write a Maple V function that defines the  $n$ th partial sum,  $s(n,x)$ , of the Fourier expansion.
4. Plot  $s(3,x)$ ,  $s(9,x)$ ,  $s(15,x)$  and  $G1(x)$  on the same graph in the interval  $[-\pi, \pi]$  and paste the plot into your worksheet.
5. Write a Maple V function,  $e(n,x)$ , that computes the error in approximating  $G1(x)$  by  $s(n,x)$ .

6. Plot the function  $e(3,x)$  on the interval  $[-\pi, \pi]$ .
7. Using the preceding plot estimate the maximum error in approximating the function  $G1$  by  $s(3,x)$  on the interval  $[-\pi, \pi]$ .
8. The plot of the function  $e(9,x)$  on the interval  $[-\pi, \pi]$ .
9. Using the plot estimate the maximum error in approximating the function  $G1$  by  $s(9,x)$  on the interval  $[-\pi, \pi]$ .
10. Using graphical methods find the smallest value of  $n$  that will give a maximum error of less than 0.04 when approximating  $G1(x)$  by  $s(n,x)$ . Justify your answer with plots.